

Excel Funktionen

Kategorie Mathematik & Trigonometrie
Kategorie Statistik



Basisfunktionen

- ▶ Folgende Funktionen werden als bekannt vorausgesetzt und nicht gesondert behandelt

Funktion	Bemerkung
=Min(...)	Kleinster Wert
=Max(...)	Grösster Wert
=Summe(...)	Addition
=Produkt(...)	Multiplikation
=Mittelwert(...)	Arithmetischer Durchschnitt
=Median(...)	Median
=Anzahl(...)	Anzahl von Zahlen
=Anzahl2(...)	Anzahl nichtleerer Zellen

...WENN und ...WENNS-Funktion

- ▶ Abwandlungen von Summe, Mittelwert und Anzahl
- ▶ Erlauben Einschränkungen, sodass die Berechnung nur für Zellen durchgeführt wird, bei denen bestimmte Kriterien erfüllt sind
- ▶ ...WENN
 - Angabe eines Suchkriteriums
- ▶ ...WENNS
 - Angabe mehrerer Suchkriterien
 - Verknüpft mit logischem UND

Suchkriterium mit literalen Konstanten

- ▶ Groß- / Kleinschreibung wird ignoriert
- ▶ Prüfen auf Gleichheit
 - Text / Datum unter Anführungszeichen (Beispiel: "Excel")
 - Zahlen ohne Anführungszeichen (Beispiel: 5)
- ▶ Prüfen auf Ungleichheit
 - Operatoren: <>, <, <=, >, >=
 - Gesamter Ausdruck unter Anführungszeichen
 - Beispiel Text: "<>Excel"
 - Beispiel Zahl: ">5"
- ▶ Textvergleich mit Platzhaltern
 - * steht für beliebig viele Zeichen
 - ? steht für ein Zeichen
 - Beispiel: "???el"
 - Beispiel: "E*el"

Suchkriterium mit Zellbezügen

- ▶ Groß-/Kleinschreibung wird ignoriert
- ▶ Prüfen auf Gleichheit
 - Angabe der entsprechenden Zelle (Beispiel: A5)
 - Kein Unterschied zwischen Text, Datum oder Zahlen
- ▶ Prüfen auf Ungleichheit
 - Operatoren: <>, <, <=, >, >=
 - Beispiel **falsch**: "<>A5"
 - A5 wird nicht als Zellbezug sondern als konstante Zeichenangabe interpretiert
 - Beispiel **richtig**: "<>" & A5
 - Verkettungsoperator & verknüpft die konstante Zeichenkette "<>" mit dem Inhalt aus Zelle A5
- ▶ Platzhalter
 - Als Teil des Zellinhaltes

Suchkriterium mit Berechnungen

- ▶ Analog Suchkriterium mit Zellbezügen
- ▶ Beispiel
 - Ihr Suchkriterium soll den Mittelwert aus den Zellen B1:B5 enthalten
 - Variante 1:
 - Sie berechnen den Mittelwert in einer eigenen Zelle (z.B. C1) und verwenden diese Zelle in Ihrem Suchkriterium
 - Variante 2:
 - Sie führen die Berechnung als Teil des Suchkriteriums aus
 - Gleichheit: Mittelwert(B1:B5)
 - Ungleichheit: "<>" & Mittelwert(B1:B5)

ZÄHLENWENN-Funktion

- ▶ Anzahl der nichtleeren Zellen eines Bereiches, deren Inhalte mit einem Suchkriterium übereinstimmen
- ▶ Zwei Argumente
 - Zellbereich und Suchkriterium

Anweisung	Anzahl der nichtleeren Zellen in A1:A10...
=ZählenWenn(A1:A10; "Herr")	...gleich Herr
=ZählenWenn(A1:A10; 10)	...gleich 10
=ZählenWenn(A1:A10; "<>Herr")	...ungleich Herr
=ZählenWenn(A1:A10; "<=10")	...kleiner oder gleich 10
=ZählenWenn(A1:A10; "?b*")	...zweites Zeichen b, beliebig lang
=ZählenWenn(A1:A10; B1)	...gleich dem Inhalt von B1
=ZählenWenn(A1:A10; "<=" & B1)	...kleiner oder gleich dem Inhalt von B1
=ZählenWenn(A1:A10; "<" & Mittelwert(B1:B5))	...kleiner als der Mittelwert von B1:B5

ZÄHLENWENNS-Funktion

- ▶ Wie ZÄHLENWENN-Funktion, erlaubt aber das Formulieren mehrerer Bereiche und Suchkriterien
 - Jeweils volle Angabe von Bereich und Suchkriterium
 - Bereiche müssen gleiche Anzahl von Zellen umfassen
 - Alle Kriterien müssen erfüllt sein (logisches UND)

Anweisung	Anzahl nichtleerer Zellen...
=ZählenWenns(A1:A10; "Herr"; B1:B10; "Schwaz")	...Werte aus A1:A10 gleich "Herr" und Werte aus B1:B10 gleich "Schwaz"
=ZählenWenns(A1:A10; ">=10"; A1:A10; "<=20")	...Werte aus A1:A10 zwischen 10 und 20 (inkl. der Grenzen)
=ZählenWenns(A1:A10; "<>" & C1; B1:B10; C2)	...Werte aus A1:A10 ungleich Inhalt von C1 und Werte aus B1:B10 gleich Inhalt von C2

SUMMEWENN-Funktion

- ▶ Addiert Werte eines Bereiches, deren Inhalte mit einem Suchkriterium übereinstimmen
- ▶ Drei Argumente
 - Bereich
 - Suchkriterium
 - Summenbereich (optional)
- ▶ Ist der Bereich auf den das Suchkriterium angewendet wird und der zu addierende Bereich ident, so kann das dritte Argument entfallen.

SUMMEWENN-Funktion

Anweisung	Bemerkung
=SummeWenn(A1:A10; 10)	Summe der Werte aus A1:A10, die gleich 10 sind
=SummeWenn(A1:A10; 10; A1:A10)	Wie oberhalb, das dritte Argument ist unnötig, da Kriterien- und Summenbereich ident sind
=SummeWenn(A1:A10; ">=10")	Summe der Werte aus A1:A10, die grösser oder gleich 10 sind
=SummeWenn(A1:A10; "Herr"; B1:B10)	Summe der Werte aus B1:B10, die in Spalte A den Eintrag "Herr" enthalten
=SummeWenn(A1:A10; "<" & Mittelwert(A1:A10))	Summe der Werte aus A1:A10, die unterhalb des Mittelwertes von A1:A10 liegen

SUMMEWENNS-Funktion

- ▶ Wie SUMMEWENN-Funktion, erlaubt aber das Formulieren mehrerer Bereiche und Suchkriterien
 - **Erstes Argument ist der Summenbereich !!!**
 - Jeweils volle Angabe von Bereich und Suchkriterium
 - Bereiche müssen gleiche Anzahl von Zellen umfassen
 - Alle Kriterien müssen erfüllt sein (logisches UND)

Anweisung	Beschreibung
=SummeWenns(A1:A10; A1:A10; ">=10"; A1:A10; "<=20")	Summe der Werte aus A1:A10, die zwischen 10 und 20 liegen (inkl. Grenzen)
=SummeWenns(A1:A10; A1:A10; "<20" ; B1:B10; "Herr)	Summe der Werte aus A1:A10, die kleiner 20 sind und in Spalte B den Eintrag "Herr" haben

MITTELWERTWENN-Funktion

- ▶ Berechnet den arithmetischen Durchschnitt der Werte eines Bereiches, deren Inhalte mit einem Suchkriterium übereinstimmen
- ▶ Drei Argumente
 - Bereich
 - Suchkriterium
 - Mittelwertbereich (optional)
- ▶ Ist der Bereich auf den das Suchkriterium angewendet wird und der zu berechnende Bereich ident, so kann das dritte Argument entfallen.

MITTELWERTWENN-Funktion

Anweisung	Bemerkung
=MittelwertWenn(A1:A10; ">=10")	Arithmetischer Durchschnitt der Werte aus A1:A10, die grösser oder gleich 10 sind
=MittelwertWenn(A1:A10; ">=10"; A1:A10)	Wie oberhalb, das dritte Argument ist unnötig, da Kriterien- und Mittelwertbereich ident sind
=MittelwertWenn(A1:A10; "Herr"; B1:B10)	Arithmetischer Durchschnitt der Werte aus B1:B10, die in Spalte A den Eintrag "Herr" enthalten
=MittelwertWenn(A1:A10; "<" & B2)	Arithmetischer Durchschnitt der Werte aus A1:A10, die unterhalb des Wertes aus B2 liegen

MITTELWERTWENNS-Funktion

- ▶ Wie MITTELWERTWENN-Funktion, erlaubt aber das Formulieren mehrerer Bereiche und Suchkriterien
 - **Erstes Argument ist der Mittelwertbereich !!!**
 - Jeweils volle Angabe von Bereich und Suchkriterium
 - Bereiche müssen gleiche Anzahl von Zellen umfassen
 - Alle Kriterien müssen erfüllt sein (logisches UND)

Anweisung	Beschreibung
=MittelwertWenns(A1:A10; A1:A10; ">=10"; A1:A10; "<=20")	Arithmetischer Durchschnitt der Werte aus A1:A10, die zwischen 10 und 20 liegen (inkl. Grenzen)
=MittelwertWenns(A1:A10"; A1:A10; "<20" ; B1:B10; "Herr")	Arithmetischer Durchschnitt der Werte aus A1:A10, die kleiner 20 sind und in Spalte B den Eintrag "Herr" haben

STABW-Funktion


- ▶ Schätzung / Berechnung der Standardabweichung
 - Durchschnittliche Streuung der Werte um den arithmetischen Mittelwert

- ▶ Varianten alt

- STABW: ausgehend von Stichprobe
- STABWN: ausgehend von Grundgesamtheit

- ▶ Varianten neu

- STABW.S: ausgehend von Stichprobe
- STABW.N: ausgehend von Grundgesamtheit



- ▶ Alte und neue Varianten gibt es zu einigen statistischen Berechnungen, verwenden Sie wenn möglich die neuen Varianten.